

$$T \cdot x = b$$

$$x_i = \frac{1}{t_{i,i}} \left(b_i - \sum_{j=i+1}^{n} t_{i,j} x_j \right)$$

```
function x = pivot_gauss(t,b)
  [n,n] = size(t); // dimension de t
  x=zeros(n,1);
  for i = n:-1:1
    y = b(i);
    for j = i+1:n
      y = y - t(i,j)*x(j);
    end
    x(i)= y/(t(i,i));
  end
endfunction
```

Pivot de Gauss (T est une matrice triangulaire de dimension $n \times n$ et b est un vecteur de taille n.)

Exemple de la méthode de Gauss

		Matrice A	Vecteur b
	Départ	$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 7 & 12 \\ 3 & 7 & 8 \end{bmatrix}$	$\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}$
Tridiagonalisation	Soustraction Ligne 1	$\begin{bmatrix} 1 & 2 & 3 \\ 0 & -1 & 0 \\ 0 & 1 & -1 \end{bmatrix}$	$\begin{bmatrix} 1 \\ -4 \\ -3 \end{bmatrix}$
	Echange Lignes 2/3	$\begin{bmatrix} 1 & 2 & 3 \\ 0 & 1 & -1 \\ 0 & -1 & 0 \end{bmatrix}$	$\begin{bmatrix} 1 \\ -3 \\ -4 \end{bmatrix}$
	Soustraction Ligne 2	$\begin{bmatrix} 1 & 2 & 3 \\ 0 & 1 & -1 \\ 0 & 0 & -1 \end{bmatrix}$	$\begin{bmatrix} 1 \\ -3 \\ -7 \end{bmatrix}$
Pivot de Gauss	Résolution Ligne 3	$\begin{bmatrix} 1 & 2 & 3 \\ 0 & 1 & -1 \\ 0 & 0 & 1 \end{bmatrix}$	$\begin{bmatrix} 1 \\ -3 \\ 7 \end{bmatrix}$
	Soustraction Ligne 3	$\begin{bmatrix} 1 & 2 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$	$\begin{bmatrix} -20 \\ 4 \\ 7 \end{bmatrix}$
	Soustraction Ligne 2	$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$	$\begin{bmatrix} -28 \\ 4 \\ 7 \end{bmatrix}$

Exemple de la méthode de Gauss

		Matrice A	Vecteur b
	Départ	$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 7 & 12 \\ 3 & 7 & 8 \end{bmatrix}$	$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$
Tridiagonalisation	Soustraction Ligne 1	$\begin{bmatrix} 1 & 2 & 3 \\ 0 & -1 & 0 \\ 0 & 1 & -1 \end{bmatrix}$	$\begin{bmatrix} 1 & 0 & 0 \\ -4 & 1 & 0 \\ -3 & 0 & 1 \end{bmatrix}$
	Echange Lignes 2/3	$\begin{bmatrix} 1 & 2 & 3 \\ 0 & 1 & -1 \\ 0 & -1 & 0 \end{bmatrix}$	$\begin{bmatrix} 1 & 0 & 0 \\ -3 & 0 & 1 \\ -4 & 1 & 0 \end{bmatrix}$
	Soustraction Ligne 2	$\begin{bmatrix} 1 & 2 & 3 \\ 0 & 1 & -1 \\ 0 & 0 & -1 \end{bmatrix}$	$\begin{bmatrix} 1 & 0 & 0 \\ -3 & 0 & 1 \\ -7 & 1 & 1 \end{bmatrix}$
Pivot de Gauss	Résolution Ligne 3	$\begin{bmatrix} 1 & 2 & 3 \\ 0 & 1 & -1 \\ 0 & 0 & 1 \end{bmatrix}$	$\begin{bmatrix} 1 & 0 & 0 \\ -3 & 0 & 1 \\ 7 & -1 & -1 \end{bmatrix}$
	Soustraction Ligne 3	$\begin{bmatrix} 1 & 2 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$	$\begin{bmatrix} -20 & 3 & 3 \\ 4 & -1 & 0 \\ 7 & -1 & -1 \end{bmatrix}$
	Soustraction Ligne 2	$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$	$\begin{bmatrix} -28 & 5 & 3 \\ 4 & -1 & 0 \\ 7 & -1 & -1 \end{bmatrix}$

```
function x = syslin_gauss(a,b)
  [n,n] = size(a);
  m = zeros(n,n+1); // Remplissage de la matrice
  m(1:n,1:n) = a;
  m(1:n,n+1) = b;

  for i = 1:n
    col = abs(m);
    [hmax,piv] = maxi(col((i:n),i)); // Pivot partiel
    piv = piv + i - 1;

    if (piv <> i) then // Echange des lignes
      m = swapl(m,i,piv,n+1);
    end

    for j = (i+1):n // Elimination des variables
      mul = m(j,i)/m(i,i);
      for k = i:(n+1);
        m(j,k)=m(j,k)-mul*m(i,k);
      end
    end
  end
  t = m(1:n,1:n);
  c = m(1:n,n+1);
  x = pivot_gauss(t,c); // Réutilisation pivot_gauss
endfunction
```

Méthode de Gauss avec stratégie du pivot partiel (a est une matrice carrée inversible de dimension $n \times n$ et b est un vecteur de taille n.)

```
function x = decomp_lu(a,b)
  [n,n] = size(a)
  p = eye(n,n);
  l = eye(n,n);
  u = a;

  for i = 1:n
    col = abs(u);          // Recherche du pivot
    [hmax,piv] = maxi(col(i:n,i));
    piv = piv + i - 1;

    if (piv <> i) then         // Echange des lignes
      u = swapl(u,i,piv,n);
      p = swapl(p,i,piv,n);
      l = swapl(l,i,piv,i);
    end

    l(i,i)=1;
    for j = (i+1):n        // Elimination des variables
      mul = u(j,i)/u(i,i);
      l(j,i) = mul;
      for k = i:n;
        u(j,k) = u(j,k)-mul*u(i,k);
      end
    end
  end
endfunction
```

Méthode de décomposition *PLU* avec stratégie du pivot partiel

```
function l = decomp_tt (a)
[n,n] = size(a)
l = zeros(n,n);
d = zeros(1,n);

for i = 1:n
  for j = i:n
    msum = a(i,j);
    for k=(i-1):-1:1
      msum = msum - l(i,k)*l(j,k);
    end
    if (i=j) then
      d(i) = sqrt(abs(msum));
      l(i,i) = d(i);
    end
    l(j,i) = msum/d(i);
  end
end
endfunction
```

Méthode de Cholesky optimisée

Méthode de Jacobi

$$A = M - N \quad \text{où} \quad \begin{cases} M = D \\ N = E + F \end{cases}$$

$$\text{Définition de la suite :} \quad \begin{cases} u_0 \text{ donné} \\ u_{n+1} = D^{-1}(E+F)u_n + D^{-1}b \end{cases}$$

$$\begin{array}{rcl}
x_1^{k+1} & = & \frac{1}{a_{1,1}} \left(\quad -a_{1,2}u_2^k \; -a_{1,3}u_3^k \quad \dots \quad -a_{1,n\text{-}1}u_{n\text{-}1}^k \; -a_{1,n}u_n^k \; + \; b_1 \right) \\
x_2^{k+1} & = & \frac{1}{a_{2,2}} \left(-a_{2,1}u_1^k \quad -a_{2,3}u_3^k \quad \dots \quad -a_{2,n\text{-}1}u_{n\text{-}1}^k \; -a_{2,n}u_n^k \; + \; b_2 \right) \\
\vdots & = & \vdots \left(\vdots \qquad\qquad\qquad\qquad\qquad \vdots \; + \; \vdots \right) \\
x_{n\text{-}1}^{k+1} & = & \frac{1}{a_{n\text{-}1,n\text{-}1}} \left(-a_{n\text{-}1,1}u_1^k - a_{n\text{-}1,2}u_2^k \quad \dots \quad -a_{n\text{-}1,n\text{-}2}u_{n\text{-}2}^k \quad -a_{n\text{-}1,n}u_n^k \; + \; b_{n\text{-}1} \right) \\
x_n^{k+1} & = & \frac{1}{a_{n,n}} \left(-a_{n,1}u_1^k \; -a_{n,2}u_2^k \quad \dots \quad -a_{n,n\text{-}2}u_{n\text{-}2}^k \; -a_{n,n\text{-}1}u_{n\text{-}1}^k \; + \; b_n \right)
\end{array}$$

Méthode de Gauss-Seidel

$$A = M - N \quad \text{où} \quad \begin{cases} M = D - E \\ N = F \end{cases}$$

$$\text{Définition de la suite :} \quad \begin{cases} u_0 \text{ donné} \\ u_{n+1} = (D-E)^{-1} F u_n + (D-E)^{-1} b \end{cases}$$

$$\begin{array}{rcllllllll}
x_1^{k+1} & = & \frac{1}{a_{1,1}} & (& & -a_{1,2}u_2^k & -a_{1,3}u_3^k & \dots & -a_{1,n-1}u_{n-1}^k & -a_{1,n}u_n^k & + & b_1 \;) \\
x_2^{k+1} & = & \frac{1}{a_{2,2}} & (& -a_{2,1}u_1^{k+1} & & -a_{2,3}u_3^k & \dots & -a_{2,n-1}u_{n-1}^k & -a_{2,n}u_n^k & + & b_2 \;) \\
\vdots & = & \vdots & (& \vdots & & & & & \vdots & + & \vdots \;) \\
x_{n-1}^{k+1} & = & \frac{1}{a_{n-1,n-1}} & (& -a_{n-1,1}u_1^{k+1} & -a_{n-1,2}u_2^{k+1} & \dots & -a_{n-1,n-2}u_{n-2}^{k+1} & & -a_{n-1,n}u_n^k & + & b_{n-1} \;) \\
x_n^{k+1} & = & \frac{1}{a_{n,n}} & (& -a_{n,1}u_1^{k+1} & -a_{n,2}u_2^{k+1} & \dots & -a_{n,n-2}u_{n-2}^{k+1} & -a_{n,n-1}u_{n-1}^{k+1} & & + & b_n \;)
\end{array}$$